
This week, tips on text. How to join text from two 
columns using the concatenation key, then, how to use 
LEFT(), MID(), FIND() and LEN() to isolate portions 
of an account ID. 

Strategy: Use the ampersand (&) as a concatenation operator in a for-
mula in column C. Change the formulas in column C to values before 
deleting columns A and B. Follow these steps.

1) In cell C2, enter the formula =A2&B2, as shown in Fig. 258.

2) You need to insert a space between the first name and last name. If 
you join cell A2, a space in quotes, and cell B2, the answer will look 
acceptable. The formula is =A2&“ ”&B2. Copy this formula down to 
all of the cells in your range, as shown in Fig. 259.

Additional Information: To convert BRITNEY SPEARS to Britney 
Spears, use the PROPER function. =PROPER(A2&“ ”&B2) will convert 
the names to Proper Case, as shown in Fig. 260. This will work for all 
of your names except names with an interior capital such as “Paul Mc-
Cartney” or “Dave VanHorn”. After using the PROPER function, you 
will have to manually fix any names with an interior capital letter.

Fig. 258

Fig. 259

Gotcha: If you delete columns A and B while column C still contains 
formulas, then all of the formulas will change to #REF! errors, as shown 
in Fig. 261. This tells you that you have a formula that points to cells(s) 
that are no longer there. Immediately hit Ctrl+Z to undo the delete.

 

To work around this situation, first convert all of the formulas in column 
C to values. Follow these steps:

1) Select the data in column C.

2) Select Ctrl+C to copy the data to the clipboard.

3) Without changing the selection, select Edit – Paste Special….

Fig. 260

Fig. 261

4) In the Paste Special dialog box, choose Values and then OK, as 
shown in Fig. 262.

5) This converts column C from live formulas to static values. You can 
now delete columns A and B.

Summary: The ampersand (&) is the concatenation character used to 
join text cells with other text cells or with literal values in a formula.

Commands Discussed: Edit – Paste Special

Functions Discussed: =PROPER()

Fig. 262

HOW TO SORT ON ONE 
SEGMENT OF AN ACCOUNT ID

Problem: Your company assigns an account ID to every customer. One 
segment of the account ID contains useful information, such as a parent 
company code. You want to be able to sort on the basis of a portion of the 
account ID. In Fig. 263, the first three digits of the account are used to 
identify an office.

Strategy: Insert a new column and use the LEFT function to isolate the 
necessary digits from the account field.

1) In the blank column, enter a heading such as the word “Key”. In 
cell G2, enter the formula =LEFT(A2,3), as shown in Fig. 264. This 
indicates that the new field should contain just the three charac-
ters leftmost of the ID field.

Fig. 263

Fig. 264

2) Double-click the Fill handle in cell G2 to copy the formula down to 
all of the rows in your dataset. The Fill handle is the black square 
dot in the lower right corner of the cell pointer.

3) Change the formulas in column G to values. 

Highlight all of the cells in column G. Use Ctrl+C to copy. As shown in 
Fig. 265, from the menu, select Edit – Paste Special – Values – OK.

Result: A certain portion of the Account field is now available in a new 
column. You can now use data tools, such as Sort, Filter, or Subtotal, to 
isolate certain offices.

Summary: When you need to isolate a portion of the characters in an-
other column, creating a temporary column is the solution.

Commands Discussed: Edit – Paste Special

Functions Discussed: =LEFT()

Fig. 265

HOW TO ISOLATE THE CENTER 
PORTION OF AN ACCOUNT ID

Problem: Your company assigns an account ID in the format of SSS-
XX-YYYY as shown in Fig. 266. You need to isolate the XX portion of the 
account ID in order to subtotal or sort the data.

Strategy: Insert a new column and use the MID function to isolate the 
necessary digits from the account field.

The MID function takes three arguments. The first argument is a cell 
containing a text value. The second argument is the character number 
where you want the result to start. The final argument is the length of 
the result.

In a well-formed account number, like 123-45-6789, you can predict that 
the start of the second segment will always be in the fifth character posi-
tion. The length of the second segment is always two characters.

Fig. 266

In a blank column, enter a heading, such as the word “Key”. In cell G2, 
enter the formula =MID(A2,5,2). Copy the formula down to all rows, as 
shown in Fig. 267.

Additional Details: In order to capture the final four digits of the account 
number, you could either use the =MID(A2,8,4) or the =RIGHT(A2,4) 
function to isolate the final four digits of the account number. 

Result: You can now sort by the new column and add subtotals by this 
field.

Summary: When you need to isolate a portion of the characters in an-
other column, creating a temporary column is the solution.

Functions Discussed: =MID(); =RIGHT()

Cross Reference: How to Sort on One Segment of an Account ID

HOW TO ISOLATE EVERYTHING BEFORE A 
DASH IN A COLUMN BY USING FUNCTIONS

Problem: A vendor provides an Excel worksheet. As shown in Fig. 268, 
one field has a manufacturer code, a dash, and a part number. The man-
ufacturer codes are not always the same length. You need to isolate the 
manufacturer code in a new column.

Fig. 267

Strategy: Use the FIND function to locate the dash in the item number. 
The FIND function will return the character position of the dash. You 
can then use that result minus one in the LEFT function to isolate the 
manufacturer code. 

The FIND function requires two arguments. The first argument is the 
text that you are trying to locate. In this case, you are trying to locate a 
dash, so you should include the dash in quotes. The second argument is 
the location of the cell containing the text to search.

1) Enter =FIND(“-”,A2) in cell F2. Copy down to all of the other cells, 
as shown in Fig. 269.

Fig. 268

Fig. 269

The 3 in cell F2 indicates that the dash is located in the third character 
position of cell A2. The 6 in cell F3 indicates that the dash is in the sixth 
position of cell A3. If you want to isolate the manufacturer code, you will 
want to grab one less than this number from the left of the code.

2) In cell G2, enter the formula =LEFT(A2,F2–1), as shown in Fig. 
270. Double-click the Fill handle to copy this formula down to all 
cells.

Alternate Strategy: You do not need to enter the formulas in two dif-
ferent columns. You could easily have used this formula in cell F2:

=LEFT(A2,FIND(“-”,A2)–1)

Additional Details: Once the formula has isolated the manufacturer 
code, change the formulas to values. See the Cross Reference.

Result: You can now sort by the new column and add subtotals by this 
field.

Summary: When you need to isolate a portion of the characters in an-
other column, creating a temporary column is the solution.

Functions Discussed: =FIND(); =LEFT()

Cross Reference: How to Sort on One Segment of an Account ID

Fig. 270

HOW TO USE FUNCTIONS TO ISOLATE 
EVERYTHING AFTER A DASH IN A COLUMN

Problem: A vendor provides an Excel worksheet. One field has a manu-
facturer code, a dash, and a part number, as shown in Fig. 271. The 
manufacturer codes are not always the same length. You need to isolate 
the part number in a new column.

Strategy: Use the MID function to extract a portion of text from the mid-
dle of the text. The MID function requires three arguments: =MID(Cell 
with Text, Character Number to Start, Number of Characters). You can 
use the FIND function to locate the dash in the item number and start 
at one character to the right. You can use the LEN function to figure out 
how long the text is.

The FIND function requires two arguments. The first argument is the 
text that you are trying to locate. In this case, you are trying to locate a 
dash, so you should include the dash in quotes. The second argument is 
the location of the cell containing the text to search.

1) Enter =FIND(“-”,A2) in cell F2. Copy down to all of the other cells, 
as shown in Fig. 272.

Fig. 271

Fig. 272

 The 3 in cell F2 indicates that the dash is located in the third char-
acter position of cell A2. If you want to isolate the part number, 
you will want to start one character to the right of this position. 
So, you know the start of your formula will be =MID(A2,F2+1. The 
hard part is figuring out how many characters there are in the part 
number.

2) In cell G2, enter the formula =LEN(A2). Copy this formula down to 
all rows, as shown in Fig. 273.

 The LEN function will tell you the total number of characters in 
cell A2. Cell G2 tells you that there are 10 characters in A2. If the 
dash is in the third position, then you want to grab (G2–F2) char-
acters or (10–3). This will give you seven characters.

3) In cell H2, enter the formula =MID(A2,F2+1,G2–F2), as shown in 
Fig. 274. In plain language, this tells Excel to extract characters 
from A2. Excel should start at the character after the result in F2 
and continue for a length of (G2–F2) characters.

Fig. 273

Fig. 274

In this case, it is much harder to enter the formula in one column instead 
of three columns. This is because the result of the FIND function is used 
twice in the H2 formula. In order to build a single formula, you will be 
entering the FIND function twice in that formula. As shown in Fig. 275, 
you could have entered the following formula in H2: =MID(A2,FIND
(“-”,A2)+1,LEN(A2)–FIND(“-”,A2)).

Alternate Strategy: You might be tempted to use a formula such 
as =MID(A2,F2+1,50). The third parameter, 50, will ensure that it 
can grab a part number of any conceivable length. This approach will 
work, but the resulting answer will have forty-four blanks appended 
to the end of the part number. This makes it impossible to use suc-
cessfully in a VLOOKUP formula as it stands. However, you can use 
=TRIM(MID(A2,F2+1,50)) to remove the trailing spaces.

Summary: Use a combination of LEN and FIND functions as an aide to 
help you locate the proper start or end position for the LEFT, MID, and 
RIGHT functions.

Functions Discussed: =FIND(); =MID(); =LEN(); =TRIM()

HOW TO USE FUNCTIONS TO ISOLATE 
EVERYTHING AFTER THE SECOND 

DASH IN A COLUMN

Problem: A vendor file contains a three-segment part number, as shown 
in Fig. 276. Each segment is separated by a dash. The length of each 
segment could be any number of characters. Find the second or third 
segment.

Fig. 275

Strategy: There is an optional third argument in the FIND function. 
This tells Excel to start looking after a certain character position in the 
text. In this case, to find the second dash, you want Excel to start look-
ing after the location of the first dash.

1) As in the prior examples, use =FIND(“-”,A2) in cell F2 to locate the 
first dash.

2) Enter =FIND(“-”,A2,F2+1) in cell G2, as shown in Fig. 277. The 
F2+1 parameter tells Excel that you want to find a dash starting in 
the fourth character position of cell A2.

3) Enter =LEFT(A2,F2–1) in H2. The formula in H2 locates the first 
segment of the part number. 

4) Enter =MID(A2,F2+1,G2–F2) in I2. The formula in I2 locates the 
middle segment of the part number. 

5) To get the right segment of the part number, you can use the RIGHT 
function. Just like the LEFT function, the RIGHT function requires 
a cell and the number of characters from the right side of the item 
number. To find the number of characters, use =LEN(A2)–G2. 

Fig. 276

Fig. 277

Enter the resulting formula, =RIGHT(A2,LEN(A2)–G2), in J2. See 
Fig. 278.

Gotcha: All of these formulas are trusting that the vendor always in-
cluded two dashes in the item number. If an item number exists without 
a second dash, the second FIND function would return a #VALUE! er-
ror, leading to errors in the calculation for the second and third items. 
Before converting formulas to values and deleting the original part num-
ber, sort the data by column F and then sort descending by column G. As 
shown in Fig. 279, any #VALUE! errors will sort to the top of the data-
set, where you can locate and correct the errors in the part number.

Summary: Using combinations of FIND, LEN, MID, LEFT, and RIGHT, 
it is possible to parse nearly any data imaginable.

Functions Discussed: =FIND(); =LEN(); =MID(); =LEFT(); =RIGHT()

Fig. 278

Fig. 279




